
"GIVEN:  An ideal Rankine Cycle with superheat as shown"
s$ = 'steam_iapws'
P[1] = 1600[psia]
T[1] = 1100[F]
P[2] = 1[psia]
P[3] = P[2]
x[3] = 0
P[4] = P[1]
T_5 = 60[F]
T_6 = 80[F]
m_dot = 1.4E6[lbm/hr]

 
"FIND: (a) The net power developed (hp)

(b) The thermal efficiency of the cycle
(c) The heat rate of the cycle
(d) The back work ratio of the cycle
(e) The mass flow rate of the cooling water (lbm/hr)"

 
"SOLUTION:"
"Start by building the property table" 
"!State 1 is defined from the given pressure and temperature."

h[1] = enthalpy(s$,P=P[1],T=T[1])
s[1] = entropy(s$,P=P[1],T=T[1])

     x[1] = quality(s$, P=P[1], T=T[1])
v[1] = volume(s$, P=P[1], T=T[1])

 
"!State 2 (Ideal) is defined from the given pressure and knowing that the entropy is the same as state 1." 

s[2] = s[1]
h[2] = enthalpy(s$,P=P[2],s=s[2])

      T[2] = temperature(s$, P=P[2], s=s[2])
      x[2] = quality(s$, P=P[2], s=s[2])

v[2] = volume(s$, P=P[2], s=s[2])
 
"!State 3 is defined from the given pressure and quality"

h[3] = enthalpy(s$,P=P[3],x=x[3])
      T[3] = temperature(s$, P=P[3], x=x[3])

s[3] = entropy(s$,P=P[3],x=x[3])
v[3] = volume(s$,P=P[3],x=x[3])

 
"!State 4 (Ideal) is defined from the given pressure and knowing that the entropy is the same as state 3." 

s[4] = s[3]
h[4] = enthalpy(s$,P=P[4],s=s[4])

     T[4] = temperature(s$, P=P[4], s=s[4])
     x[4] = quality(s$, P=P[4], s=s[4])

v[4] = volume(s$, P=P[4], s=s[4])
 
"Now that property table is built, apply First Law to each component"
 "Turbine"

W_dot_t*convert(hp,Btu/hr) = m_dot*(h[1] - h[2])
"Boiler"
Q_dot_in = m_dot*(h[1] - h[4])
"Pump,"
W_dot_p*convert(hp,Btu/hr) = m_dot*(h[4] - h[3]) 
"Condensor"

 Q_dot_out = m_dot*(h[2] - h[3])
 
 
"!Calculating Parameters for the Rankine Cycle with Superheat"
 
"The thermal efficiency of the cycle is defined as,"

eta_th = W_dot_net*convert(hp,Btu/hr)/Q_dot_in
 
"The heat rate (HR) is defined as,"

HR = Q_dot_in/(W_dot_net*convert(hp,kW))

"The back work ratio (bwr) is defined as,"

bwr = W_dot_p/W_dot_t
 
"Mass flow rate of the cooling water"
"To determine the cooling water flow rate, draw a system boundary
 that keeps the heat transferred between the condensing steam and
 the cooling water all internal.  The First Law applied to this system is,"

Q_dot_out = m_dot_w*cp_w*(T_6 - T_5)
 
"The heat capacity of the liquid can be estimated as the saturated liquid
 value at the average temperature"

T_avg = (T_5 + T_6)/2
cp_w = cp(s$,T=T_avg,x=0)

 
"!Net power delivery from the cycle"
"The net power developed by the cycle is,"

W_dot_net = W_dot_t - W_dot_p
 
 
"Calculate the Carnot Efficiency,"

eta_Carnot = 1 - T_L/T_H
T_L = converttemp(F,R,T_sat(s$,P=P[2]))
T_H = converttemp(F,R,T[1])

SOLUTION

Unit Settings: Eng F psia mass deg

bwr  = 0.007483 [dim] cpw  = 0.9992 [Btu/lbm-R]

ηCarnot = 0.6401 [dim] ηth  = 0.4288 [dim]

HR  = 7957 [Btu/kW-hr] m = 1.400E+06 [lbm/hr]

mw = 5.896E+07 [lbm/hr] Qin = 2.063E+09 [Btu/hr]

Qout = 1.178E+09 [Btu/hr] s$  = 'steam_iapws' 

T5  = 60 [F] T6  = 80 [F]

Tavg = 70 [F] TH  = 1560 [R]

TL  = 561.4 [R] Wnet  = 347635 [hp] {8.845E+08 [Btu/hr]}

Wp  = 2621 [hp] {6.669E+06 [Btu/hr]} W t  = 350256 [hp] {8.912E+08 [Btu/hr]}

No unit problems were detected.

KEY VARIABLES

W t  = 350256 [hp] {8.912E+08 [Btu/hr]} Total turbine power delivery

m = 1.400E+06 [lbm/hr] Mass flow rate of steam in the cycle

Wp  = 2621 [hp] {6.669E+06 [Btu/hr]} Pump power required

Wnet  = 347635 [hp] {8.845E+08 [Btu/hr]} Net power delivery from the cycle

Qin = 2.063E+09 [Btu/hr] Heat transfer rate at the boiler

ηth  = 0.4288 [dim] Thermal efficiency of the cycle

HR  = 7957 [Btu/kW-hr] Heat rate of the cycle

bwr  = 0.007483 [dim] Back work ratio of the cycle

mw = 5.896E+07 [lbm/hr] Mass flow rate of cooling water



File:Example 1 - Ideal Rankine Cycle with Superheat.EES 3/31/2023 11:06:38 AM  Page 4

EES Ver. 10.836: #2191: For use only by students and faculty in Mechanical Engineering, Univ. of Idaho, Moscow, Idaho

Arrays Table: Main

Pi Ti hi si xi vi

[psia] [F] [Btu/lbm] [Btu/lbm-F] [ft3/lbm]

 1  1600  1100  1548  1.632  100  0.549 

 2  1  101.7  911.4  1.632  0.8126  271 

 3  1  101.7  69.72  0.1326  0  0.01614 

 4  1600  102.3  74.49  0.1326  -100  0.01606 

